NAG Fortran Library Routine Document

D02QFF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D02QFF is a routine for integrating a non-stiff system of first-order ordinary differential equations using a variable-order variable-step Adams method. A root-finding facility is provided.

2 Specification

```
SUBROUTINE D02QFF(FCN, NEQF, T, Y, TOUT, G, NEQG, ROOT, RWORK, LRWORK,
1                 IWORK, LIWORK, IFAIL)
INTEGER           NEQF, NEQG, LRWORK, IWORK(LIWORK), LIWORK, IFAIL
real              T, Y(NEQF), TOUT, G, RWORK(LRWORK)
LOGICAL           ROOT
EXTERNAL          FCN, G
```

3 Description

Given the initial values $x, y_1, y_2, \ldots, y_{\mathrm{NEQF}}$ the routine integrates a non-stiff system of first-order differential equations of the type, $y_i' = f_i(x, y_1, y_2, \ldots, y_{\mathrm{NEQF}})$, for $i = 1, 2, \ldots, \mathrm{NEQF}$, from $x = \mathrm{T}$ to $x = \mathrm{TOUT}$ using a variable-order variable-step Adams method. The system is defined by a subroutine FCN supplied by the user, which evaluates f_i in terms of x and $y_1, y_2, \ldots, y_{\mathrm{NEQF}}$, and $y_1, y_2, \ldots, y_{\mathrm{NEQF}}$ are supplied at $x = \mathrm{T}$. The routine is capable of finding roots (values of x) of prescribed event functions of the form

$$g_j(x, y, y') = 0, \quad j = 1, 2, \ldots, \mathrm{NEQG}.$$

Each g_j is considered to be independent of the others so that roots are sought of each g_j individually. The root reported by the routine will be the first root encountered by any g_j. Two techniques for determining the presence of a root in an integration step are available: the sophisticated method described in Watts (1985) and a simplified method whereby sign changes in each g_j are looked for at the ends of each integration step. The event functions are defined by a real function G supplied by the user which evaluates g_j in terms of $x, y_1, \ldots, y_{\mathrm{NEQF}}$ and $y_1', \ldots, y_{\mathrm{NEQF}}'$. In one-step mode the routine returns an approximation to the solution at each integration point. In interval mode this value is returned at the end of the integration range. If a root is detected this approximation is given at the root. The user selects the mode of operation, the error control, the root-finding technique and various optional inputs by a prior call of the setup routine D02QWF.

For a description of the practical implementation of an Adams formula see Shampine and Gordon (1975) and Shampine and Watts (1979).

4 References

Shampine L F and Gordon M K (1975) *Computer Solution of Ordinary Differential Equations – The Initial Value Problem* W H Freeman & Co., San Francisco

Shampine L F and Watts H A (1979) DEPAC – design of a user oriented package of ODE solvers *Report SAND79-2374* Sandia National Laboratory

Watts H A (1985) RDEAM – An Adams ODE code with root solving capability *Report SAND85-1595* Sandia National Laboratory

5 Parameters

1: FCN – SUBROUTINE, supplied by the user. *External Procedure*

FCN must evaluate the functions f_i (that is the first derivatives y_i') for given values of its arguments x, $y_1, y_2, \ldots, y_{\mathrm{NEQF}}$.

Its specification is:

```
SUBROUTINE FCN(NEQF, X, Y, F)
INTEGER          NEQF
real             X, Y(NEQF), F(NEQF)
```

1: NEQF – INTEGER *Input*

On entry: the number of differential equations.

2: X – ***real*** *Input*

On entry: the current value of the argument x.

3: Y(NEQF) – ***real*** array *Input*

On entry: the current value of the argument y_i, for $i = 1, 2, \ldots, \mathrm{NEQF}$.

4: F(NEQF) – ***real*** array *Output*

On exit: the value of f_i, for $i = 1, 2, \ldots, \mathrm{NEQF}$.

FCN must be declared as EXTERNAL in the (sub)program from which D02QFF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

2: NEQF – INTEGER *Input*

On entry: the number of first-order ordinary differential equations to be solved by D02QFF. It must contain the same value as the parameter NEQF used in a prior call of D02QWF.

Constraint: NEQF ≥ 1.

3: T – ***real*** *Input/Output*

On entry: after a call to D02QWF with STATEF = 'S' (i.e., an initial entry), T must be set to the initial value of the independent variable x.

On exit: the value of x at which y has been computed. This may be an intermediate output point, a root, TOUT or a point at which an error has occurred. If the integration is to be continued, possibly with a new value for TOUT, T must not be changed.

4: Y(NEQF) – ***real*** array *Input/Output*

On entry: the initial values of the solution $y_1, y_2, \ldots, y_{\mathrm{NEQF}}$.

On exit: the computed values of the solution at the exit value of T. If the integration is to be continued, possibly with a new value for TOUT, these values must not be changed.

5: TOUT – ***real*** *Input*

On entry: the next value of x at which a computed solution is required. For the initial T, the input value of TOUT is used to determine the direction of integration. Integration is permitted in either direction. If TOUT = T on exit, TOUT must be reset beyond T **in the direction of integration**, before any continuation call.

6: G – ***real*** FUNCTION, supplied by the user. *External Procedure*

G must evaluate a given component of $g(x, y, y')$ at a specified point.

If root-finding is not required the actual argument for G must be the dummy routine D02QFZ. (D02QFZ is included in the NAG Fortran Library and so need not be supplied by the user. Its name may be implementation dependent: see the Users' Note for your implementation for details.)

Its specification is:

```
real FUNCTION G(NEQF, X, Y, YP, K)
INTEGER          NEQF, K
real             X, Y(NEQF), YP(NEQF)
```

1: NEQF – INTEGER *Input*

On entry: the number of differential equations being solved.

2: X – ***real*** *Input*

On entry: the current value of the independent variable.

3: Y(NEQF) – ***real*** array *Input*

On entry: the current values of the dependent variables.

4: YP(NEQF) – ***real*** array *Input*

On entry: the current values of the derivatives of the dependent variables.

5: K – INTEGER *Input*

On entry: the component of g which must be evaluated.

G must be declared as EXTERNAL in the (sub)program from which D02QFF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

7: NEQG – INTEGER *Input*

On entry: the number of event functions which the user is defining for root-finding. If root-finding is not required the value for NEQG must be ≤ 0. Otherwise it must be the same parameter NEQG used in the prior call to D02QWF.

8: ROOT – LOGICAL *Output*

On exit: if root-finding was required (NEQG > 0 on entry), then ROOT specifies whether or not the output value of the parameter T is a root of one of the event functions. If ROOT = .FALSE., then no root was detected, whereas ROOT = .TRUE. indicates a root and the user should make a call to D02QYF for further information.

If root-finding was not required (NEQG $= 0$ on entry) then on exit ROOT = .FALSE..

9: RWORK(LRWORK) – ***real*** array *Workspace*

This **must** be the same parameter RWORK as supplied to D02QWF. It is used to pass information from D02QWF to D02QFF, and from D02QFF to D02QXF, D02QYF and D02QZF. Therefore the contents of this array **must not** be changed before the call to D02QFF or calling any of the routines D02QXF, D02QYF and D02QZF.

10: LRWORK – INTEGER *Input*

On entry: the dimension of the array RWORK as declared in the (sub)program from which D02QFF is called.

This must be the same parameter LRWORK as supplied to D02QWF.

11: IWORK(LIWORK) – INTEGER array *Workspace*

This **must** be the same parameter IWORK as supplied to D02QWF. It is used to pass information from D02QWF to D02QFF, and from D02QFF to D02QXF, D02QYF and D02QZF. Therefore the contents of this array **must not** be changed before the call to D02QFF or calling any of the routines D02QXF, D02QYF and D02QZF.

12: LIWORK – INTEGER *Input*

On entry: the dimension of the array IWORK as declared in the (sub)program from which D02QFF is called.

This must be the same parameter LIWORK as supplied to D02QWF.

13: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, the integrator detected an illegal input, or D02QWF has not been called prior to the call to the integrator. If on entry IFAIL $= 0$ or -1, the form of the error will be detailed on the current error message unit (as defined by X04AAF).

This error may be caused by overwriting elements of RWORK and IWORK.

IFAIL $= 2$

The maximum number of steps has been attempted (at a cost of about 2 calls to FCN per step). (See parameter MAXSTP in D02QWF.) If integration is to be continued then the user need only reset IFAIL and call the routine again and a further MAXSTP steps will be attempted.

IFAIL $= 3$

The step size needed to satisfy the error requirements is too small for the ***machine precision*** being used. (See parameter TOLFAC in D02QXF.)

IFAIL $= 4$

Some error weight w_i became zero during the integration (see parameters VECTOL, RTOL and ATOL in D02QWF.) Pure relative error control (ATOL $= 0.0$) was requested on a variable (the ith) which has now become zero. (See parameter BADCMP in D02QXF.) The integration was successful as far as T.

IFAIL $= 5$

The problem appears to be stiff (see Chapter D02 for a discussion of the term 'stiff'). Although it is inefficient to use this integrator to solve stiff problems, integration may be continued by resetting IFAIL and calling the routine again.

IFAIL = 6

A change in sign of an event function has been detected but the root-finding process appears to have converged to a singular point T rather than a root. Integration may be continued by resetting IFAIL and calling the routine again.

IFAIL = 7

The code has detected two successive error exits at the current value of T and cannot proceed. Check all input variables.

7 Accuracy

The accuracy of integration is determined by the parameters VECTOL, RTOL and ATOL in a prior call to D02QWF. Note that only the local error at each step is controlled by these parameters. The error estimates obtained are not strict bounds but are usually reliable over one step. Over a number of steps the overall error may accumulate in various ways, depending on the properties of the differential equation system. The code is designed so that a reduction in the tolerances should lead to an approximately proportional reduction in the error. The user is strongly recommended to call D02QFF with more than one set of tolerances and to compare the results obtained to estimate their accuracy.

The accuracy obtained depends on the type of error test used. If the solution oscillates around zero a relative error test should be avoided, whereas if the solution is exponentially increasing an absolute error test should not be used. If different accuracies are required for different components of the solution then a component-wise error test should be used. For a description of the error test see the specifications of the parameters VECTOL, ATOL and RTOL in the routine document for D02QWF.

The accuracy of any roots located will depend on the accuracy of integration and may also be restricted by the numerical properties of $g(x, y, y')$. When evaluating g the user should try to write the code so that unnecessary cancellation errors will be avoided.

8 Further Comments

If the routine fails with IFAIL = 3 then the combination of ATOL and RTOL may be so small that a solution cannot be obtained, in which case the routine should be called again with larger values for RTOL and/or ATOL. If the accuracy requested is really needed then the user should consider whether there is a more fundamental difficulty. For example:

(a) in the region of a singularity the solution components will usually be of a large magnitude. The routine could be used in one-step mode to monitor the size of the solution with the aim of trapping the solution before the singularity. In any case numerical integration cannot be continued through a singularity, and analytical treatment may be necessary;

(b) for 'stiff' equations, where the solution contains rapidly decaying components, the routine will require a very small step size to preserve stability. This will usually be exhibited by excessive computing time and sometimes an error exit with IFAIL = 3, but usually an error exit with IFAIL = 2 or 5. The Adams methods are not efficient in such cases and the user should consider using a routine from the sub-chapter D02M–D02N. A high proportion of failed steps (see parameter NFAIL in D02QXF) may indicate stiffness but there may be other reasons for this phenomenon.

D02QFF can be used for producing results at short intervals (for example, for graph plotting); the user should set CRIT = .TRUE. and TCRIT to the last output point required in a prior call to D02QWF and then set TOUT appropriately for each output point in turn in the call to D02QFF.

9 Example

We solve the equation

$$y'' = -y, \quad y(0) = 0, \quad y'(0) = 1$$

reposed as

$$y_1' = y_2$$

$$y_2' = -y_1$$

over the range $[0, 10.0]$ with initial conditions $y_1 = 0.0$ and $y_2 = 1.0$ using vector error control (VECTOL = .TRUE.) and computation of the solution at TOUT = 10.0 with TCRIT = 10.0 (CRIT = .TRUE.). Also, we use D02QFF to locate the positions where $y_1 = 0.0$ or where the first component has a turning point, that is $y_1' = 0.0$.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     D02QFF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          NEQF, NEQG, LATOL, LRTOL, LRWORK, LIWORK
      PARAMETER        (NEQF=2,NEQG=2,LATOL=NEQF,LRTOL=NEQF,
     +                 LRWORK=23+23*NEQF+14*NEQG,LIWORK=21+4*NEQG)
      real             TSTART, HMAX
      PARAMETER        (TSTART=0.0e0,HMAX=0.0e0)
*     .. Local Scalars ..
      real             HLAST, HNEXT, T, TCRIT, TCURR, TOLFAC, TOUT
      INTEGER          BADCMP, I, IFAIL, INDEX, MAXSTP, NFAIL, NSUCC,
     +                 ODLAST, ODNEXT, TYPE
      LOGICAL          ALTERG, CRIT, ONESTP, ROOT, SOPHST, VECTOL
      CHARACTER*1      STATEF
*     .. Local Arrays ..
      real             ATOL(LATOL), RESIDS(NEQG), RTOL(LRTOL),
     +                 RWORK(LRWORK), Y(NEQF), YP(NEQF)
      INTEGER          EVENTS(NEQG), IWORK(LIWORK)
*     .. External Functions ..
      real             GTRY02
      EXTERNAL         GTRY02
*     .. External Subroutines ..
      EXTERNAL         D02QFF, D02QWF, D02QXF, D02QYF, FTRY02
*     .. Executable Statements ..
      WRITE (NOUT,*) 'D02QFF Example Program Results'
      TCRIT = 10.0e0
      STATEF = 'S'
      VECTOL = .TRUE.
      ONESTP = .FALSE.
      CRIT = .TRUE.
      MAXSTP = 0
      SOPHST = .TRUE.
      DO 20 I = 1, NEQF
         RTOL(I) = 1.0e-4
         ATOL(I) = 1.0e-6
   20 CONTINUE
      IFAIL = 0
*
      CALL D02QWF(STATEF,NEQF,VECTOL,ATOL,LATOL,RTOL,LRTOL,ONESTP,CRIT,
     +            TCRIT,HMAX,MAXSTP,NEQG,ALTERG,SOPHST,RWORK,LRWORK,
     +            IWORK,LIWORK,IFAIL)
*
      T = TSTART
      TOUT = TCRIT
      Y(1) = 0.0e0
      Y(2) = 1.0e0
*
   40 IFAIL = -1
*
      CALL D02QFF(FTRY02,NEQF,T,Y,TOUT,GTRY02,NEQG,ROOT,RWORK,LRWORK,
```

```
     +            IWORK,LIWORK,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
*
         CALL D02QXF(NEQF,YP,TCURR,HLAST,HNEXT,ODLAST,ODNEXT,NSUCC,
     +              NFAIL,TOLFAC,BADCMP,RWORK,LRWORK,IWORK,LIWORK,
     +              IFAIL)
*
         IF (ROOT) THEN
*
            CALL D02QYF(NEQG,INDEX,TYPE,EVENTS,RESIDS,RWORK,LRWORK,
     +                  IWORK,LIWORK,IFAIL)
*
            WRITE (NOUT,*)
            WRITE (NOUT,99999) 'Root at ', T
            WRITE (NOUT,99998) 'for event equation ', INDEX,
     +        ' with type', TYPE, ' and residual ', RESIDS(INDEX)
            WRITE (NOUT,99999) ' Y(1) = ', Y(1), '   Y''(1) = ', YP(1)
            DO 60 I = 1, NEQG
               IF (I.NE.INDEX) THEN
                  IF (EVENTS(I).NE.0) THEN
                     WRITE (NOUT,99998) 'and also for event equation ',
     +                 I, ' with type', EVENTS(I), ' and residual ',
     +                 RESIDS(I)
                  END IF
               END IF
   60       CONTINUE
            IF (TCURR.LT.TOUT) GO TO 40
         END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,1P,e13.5,A,1P,e13.5)
99998 FORMAT (1X,A,I2,A,I3,A,1P,e13.5)
      END
*
      SUBROUTINE FTRY02(NEQF,T,Y,YP)
*     .. Scalar Arguments ..
      real               T
      INTEGER            NEQF
*     .. Array Arguments ..
      real               Y(NEQF), YP(NEQF)
*     .. Executable Statements ..
      YP(1) = Y(2)
      YP(2) = -Y(1)
      RETURN
      END
*
      real  FUNCTION GTRY02(NEQF,T,Y,YP,K)
*     .. Scalar Arguments ..
      real                T
      INTEGER             K, NEQF
*     .. Array Arguments ..
      real                Y(NEQF), YP(NEQF)
*     .. Executable Statements ..
      IF (K.EQ.1) THEN
         GTRY02 = YP(1)
      ELSE
         GTRY02 = Y(1)
      END IF
      RETURN
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 D02QFF Example Program Results

 Root at   0.00000E+00
 for event equation  2 with type  1 and residual   0.00000E+00
  Y(1) =   0.00000E+00   Y'(1) =   1.00000E+00

 Root at   1.57076E+00
 for event equation  1 with type  1 and residual  -5.90726E-16
  Y(1) =   1.00003E+00   Y'(1) =  -5.90726E-16

 Root at   3.14151E+00
 for event equation  2 with type  4 and residual  -1.28281E-16
  Y(1) =  -1.28281E-16   Y'(1) =  -1.00012E+00

 Root at   4.71228E+00
 for event equation  1 with type  4 and residual   3.59623E-16
  Y(1) =  -1.00010E+00   Y'(1) =   3.59623E-16

 Root at   6.28306E+00
 for event equation  2 with type  1 and residual   2.47333E-15
  Y(1) =   2.47333E-15   Y'(1) =   9.99979E-01

 Root at   7.85379E+00
 for event equation  1 with type  1 and residual  -3.20716E-15
  Y(1) =   9.99970E-01   Y'(1) =  -3.20716E-15

 Root at   9.42469E+00
 for event equation  2 with type  1 and residual  -2.90637E-15
  Y(1) =  -2.90637E-15   Y'(1) =  -9.99854E-01
```